
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 12.12 which is part of the ‘IMS_34229_IWD_13wk24’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk24’.

3. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk24’ release.
Change 1 - cr_SDP_Attribute_inactive
	Function name
	cr_SDP_Attribute_inactive

	Reason for change
	Inactive is a SDP attribute with no value

	Summary of change
	Delete mandatory value and replace by empty set

	TTCN module
	Common/IMS/IMS_SDP_Templates

	MCC160 Comment
	Accepted as improvement


Before change

template (present) SDP_attribute cr_SDP_Attribute_inactive :=
  { /* @status    APPROVED_IMS */

    inactive := ? 
  };

After change

template (present) SDP_attribute cr_SDP_Attribute_inactive :=
  { /* @status    APPROVED_IMS */

    inactive := {} 
  };

Change 2 - cr_SDP_Attribute_sendrecv
	Function name
	cr_SDP_Attribute_sendrecv

	Reason for change
	Sendrecv is a SDP attribute with no value

	Summary of change
	Delete mandatory value and replace by empty set

	TTCN module
	Common/IMS/IMS_SDP_Templates

	MCC160 Comment
	Accepted as improvement


Before change

template (present) SDP_attribute cr_SDP_Attribute_sendrecv :=
  { /* @status    APPROVED_IMS */

    sendrecv := ? 
  };

After change

template (present) SDP_attribute cr_SDP_Attribute_sendrecv :=
  { /* @status    APPROVED_IMS */

    sendrecv := {} 
  };

Change 3 – IMS_CommonDefs
	Module name
	IMS_CommonDefs

	Reason for change
	Template to check to-tag uses constant tsc_IMS_InviteToTag. This constant is not in line with the generation of to tag in function f_IMS_GenerateTag

	Summary of change
	Change constant accordingly

	TTCN module
	Common/IMS/IMS_CommonDefs

	MCC160 Comment
	See R5s130497 change 1


Before change

module IMS_CommonDefs {
…

  const charstring tsc_IMS_InviteToTag           := "abc-InviteToTag"; 
After change

module IMS_CommonDefs {
…
  const charstring tsc_IMS_InviteToTag           :=  "abc-INVITE-Totag"; 
Change 4 - f_IMS_BuildSDP_AnnexC21_Step5or7
	Function name
	f_IMS_BuildSDP_AnnexC21_Step5or7

	Reason for change
	fmtp lines are to provide parameters specific to a particular format (RFC4566). 

	Summary of change
	Specify a meaningful value or a wildcard. As 34.229-1, C.21, step 5 is inconclusive (it just lists “a=fmtp: (format”)which seems to indicate just a type number and nothing else), the submitter understands this to mean an arbitrary value. Therefore, wildcard is assumed.

	TTCN module
	Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	Accepted


Before change

function f_IMS_BuildSDP_AnnexC21_Step5or7(charstring p_Fmt) runs on IMS_PTC return template (present) SDP_Message

  {

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_Fmt, tsc_CODEC_AMR_8000_1),        

      cr_SDP_Attribute_fmtp(p_Fmt),                                

      cr_SDP_Attribute_sendrecv,

      // precondition attributes:

      cr_SDP_Attribute_curr_quos(c_local, c_sendrecv),

      cr_SDP_Attribute_curr_quos(c_remote),

      cr_SDP_Attribute_des_quos(c_mandatory, c_local),

      cr_SDP_Attribute_des_quos(c_optional, c_remote),

      *

    };

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(p_Fmt),            // @sic R5-123505 sic@

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
After change

function f_IMS_BuildSDP_AnnexC21_Step5or7(charstring p_Fmt) runs on IMS_PTC return template (present) SDP_Message
  {

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_Fmt, tsc_CODEC_AMR_8000_1),       

      cr_SDP_Attribute_fmtp(p_Fmt, {“*”}),    

      cr_SDP_Attribute_sendrecv,

      // precondition attributes:

      cr_SDP_Attribute_curr_quos(c_local, c_sendrecv),

      cr_SDP_Attribute_curr_quos(c_remote),

      cr_SDP_Attribute_des_quos(c_mandatory, c_local),

      cr_SDP_Attribute_des_quos(c_optional, c_remote),

      *

    };

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(p_Fmt),            // @sic R5-123505 sic@

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
Change 5 - f_IMS_PrackRequest_MessageHeaderRX
	Function name
	f_IMS_PrackRequest_MessageHeaderRX

	Reason for change
	1. PRACK is its own message with its own Cseq number which is to be incremented compared to the preceding message in the dialog, rather than to increment it relative to the preceding reliable response – such a preceding reliable response might come late in the flow (see for instance the 180 Ringing) and still carry the Cseq of the INVITE. 
2. PRACK is its own message and carries its own Via header. This Via header does not depend on the Via header of a preceding reliable response. Especially, it can use a transport different from the preceding reliable response.

	Summary of change
	1. Use existing function f_IMS_PTC_ImsInfo_CseqIncr to calculate the Cseq number for PRACK.
2. Remove constraints on Via header – at a minimum, the constraints on the transport. Constraints on the rest of the Via header are up for discussion (see also Change 6 for a similar issue where we propose an alternative solution)

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	1. accepted, see R5s130497 change 3; R5-133188
2. Check of the Via header cannot just be removed as Via header is included in 34.229-1 A.2.4 ( fl_Via_AsInPrevInvite is changed to allow “TCP” or “UDP” as transport protocol


Before change

function f_IMS_PrackRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                              template (value) MessageHeader p_MessageHeader_PrevReliableResponse,

                                              template ContentType p_ContentType := *) runs on IMS_PTC return template (present) MessageHeader

  { /* p_MessageHeader_PrevReliableResponse   ..   MessageHeader of "the reliable response" (e.g. previous 183 response) */

    var MessageHeader v_MessageHeader_PrevReliableResponse := valueof(p_MessageHeader_PrevReliableResponse);

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Prack := cr_MessageHeader_Dummy;

    var CSeq v_CSeq := v_MessageHeader_PrevReliableResponse.cSeq;   

    var integer v_CSeqValue := v_CSeq.seqNumber + 1; 
    var integer v_ResponseNumber := v_MessageHeader_PrevReliableResponse.rSeq.responseNum;   /* @sic R5s130333 change 5.2 sic@ */

    f_IMS_PTC_ImsInfo_CseqSet(dialogMO, v_CSeqValue);

    v_MessageHeader_Prack.via             := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); // @sic R5s120530, change 3.9 sic@   
    v_MessageHeader_Prack.route           := fl_Route_ReverseOrderOfRecordRoute(v_MessageHeader_PrevReliableResponse);

    v_MessageHeader_Prack.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!

    v_MessageHeader_Prack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Prack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Prack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Prack.cSeq            := cr_CseqDef(v_CSeqValue, "PRACK");

    v_MessageHeader_Prack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Prack.rAck            := cr_RAckDef(v_ResponseNumber, v_CSeq.seqNumber, v_CSeq.method);
    v_MessageHeader_Prack.contentType     := p_ContentType;

    return v_MessageHeader_Prack;

  }
After change

function f_IMS_PrackRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,
                                              template (value) MessageHeader p_MessageHeader_PrevReliableResponse,

                                              template ContentType p_ContentType := *) runs on IMS_PTC return template (present) MessageHeader

  { /* p_MessageHeader_PrevReliableResponse   ..   MessageHeader of "the reliable response" (e.g. previous 183 response) */

    var MessageHeader v_MessageHeader_PrevReliableResponse := valueof(p_MessageHeader_PrevReliableResponse);

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Prack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMO); 
    var integer v_ResponseNumber := v_MessageHeader_PrevReliableResponse.rSeq.responseNum;   /* @sic R5s130333 change 5.2 sic@ */

    f_IMS_PTC_ImsInfo_CseqSet(dialogMO, v_CSeqValue);

    v_MessageHeader_Prack.route           := fl_Route_ReverseOrderOfRecordRoute(v_MessageHeader_PrevReliableResponse);

    v_MessageHeader_Prack.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!

    v_MessageHeader_Prack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Prack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Prack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Prack.cSeq            := cr_CseqDef(v_CSeqValue, "PRACK");

    v_MessageHeader_Prack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Prack.rAck            := cr_RAckDef(v_ResponseNumber, v_CSeq.seqNumber, v_CSeq.method);
    v_MessageHeader_Prack.contentType     := p_ContentType;

    return v_MessageHeader_Prack;

  }
Change 6 - f_IMS_AckRequest_MessageHeaderRX
	Function name
	f_IMS_AckRequest_MessageHeaderRX

	Reason for change
	According to RFC 3261, Section 17, ACK is not part of the INVITE transaction if the final response being acked was a 2xx. It constitutes its own separate transaction (also see Section 6 of RFC3261, SIP Transaction).
Also note that the header fields of an ACK are constructed in the same way as for any request within a dialog (RCF3261, 13.2.2.4). Only select header fields are dependent on the INVITE it acks.

This implies that ACK has its own Via header field. Especially, the sent-protocol in the Via header field of ACK does not depend on the response being acked, see 34.229-1, A.2.7


	Summary of change
	Added a statement to change the  Via header such that either transport can be used

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	Accepted in principle: fl_Via_AsInPrevInvite changed instead (see change 5.2)


Before change

function f_IMS_AckRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,
                                            template (omit) StatusLine p_StatusLine_PrevResponse := omit,

                                            template (omit) MessageHeader p_MessageHeader_PrevResponse := omit) return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Ack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);

    var integer v_StatusCode;

    if (ispresent(p_StatusLine_PrevResponse)) {

      v_StatusCode := valueof(p_StatusLine_PrevResponse.statusCode);

      if ((v_StatusCode < 200) or (v_StatusCode > 299)) {                                     /* @sic R5-130682: ACK for non-2xx response sic@ */

        v_MessageHeader_Ack.via         := v_MessageHeader_Invite.via;

      } else {

        v_MessageHeader_Ack.via         := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); /* @sic R5-130682: ACK for 2xx response sic@ */

      }

    }

    v_MessageHeader_Ack.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_PrevResponse);

    v_MessageHeader_Ack.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!

    v_MessageHeader_Ack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Ack.cSeq            := cr_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Ack.pAccessNetworkInfo := omit;

    return v_MessageHeader_Ack;

  }
After change

function f_IMS_AckRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,
                                            template (omit) StatusLine p_StatusLine_PrevResponse := omit,

                                            template (omit) MessageHeader p_MessageHeader_PrevResponse := omit) return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Ack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);

    var integer v_StatusCode;

    if (ispresent(p_StatusLine_PrevResponse)) {

      v_StatusCode := valueof(p_StatusLine_PrevResponse.statusCode);

      if ((v_StatusCode < 200) or (v_StatusCode > 299)) {                                     /* @sic R5-130682: ACK for non-2xx response sic@ */

        v_MessageHeader_Ack.via         := v_MessageHeader_Invite.via;

      } else {

        v_MessageHeader_Ack.via         := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); /* @sic R5-130682: ACK for 2xx response sic@ */

        v_MessageHeader_Ack.via.viaBody[0].sentProtocol.transport := ?; 
      }

    }

    v_MessageHeader_Ack.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_PrevResponse);

    v_MessageHeader_Ack.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!

    v_MessageHeader_Ack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Ack.cSeq            := cr_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Ack.pAccessNetworkInfo := omit;

    return v_MessageHeader_Ack;

  }
Change 7 - f_IMS_ByeRequest_MessageHeaderRX
	Function name
	f_IMS_ByeRequest_MessageHeaderRX

	Reason for change
	UE can pick any transport for BYE

	Summary of change
	Changed Via header such that either transport can be used

Note: in general, it might be worthwhile to go over all usages of fl_Via_AsInPrevInvite and double-check if it is appropriate to just copy the Via header from the Invite

	TTCN module
	Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	Accepted in principle: fl_Via_AsInPrevInvite changed instead (see change 5.2)


Before change

  function f_IMS_ByeRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) MessageHeader p_MessageHeader_Prev183Response := omit,

                                            boolean p_IsGiba := false) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Bye := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMO);

    v_MessageHeader_Bye.via             := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via);

    v_MessageHeader_Bye.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_Prev183Response);

    v_MessageHeader_Bye.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!

    v_MessageHeader_Bye.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Bye.cSeq            := cr_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cr_MaxForwardsDef;

    if (not p_IsGiba) {

      v_MessageHeader_Bye.require            := cr_Require;

      v_MessageHeader_Bye.proxyRequire       := cr_ProxyRequireDef;

      v_MessageHeader_Bye.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Bye.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    } else {

      v_MessageHeader_Bye.securityVerify     := omit;

    }
After change

  function f_IMS_ByeRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) MessageHeader p_MessageHeader_Prev183Response := omit,

                                            boolean p_IsGiba := false) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Bye := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMO);

    v_MessageHeader_Bye.via             := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via);

    v_MessageHeader_Bye.via.viaBody[0].sentProtocol.transport := ?;  

    v_MessageHeader_Bye.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_Prev183Response);

    v_MessageHeader_Bye.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!

    v_MessageHeader_Bye.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Bye.cSeq            := cr_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cr_MaxForwardsDef;

    if (not p_IsGiba) {

      v_MessageHeader_Bye.require            := cr_Require;

      v_MessageHeader_Bye.proxyRequire       := cr_ProxyRequireDef;

      v_MessageHeader_Bye.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Bye.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    } else {

      v_MessageHeader_Bye.securityVerify     := omit;

    }
Change 8 - f_IMS_OtherResponse_200_MessageHeaderTX
	Function name
	f_IMS_OtherResponse_200_MessageHeaderTX

	Reason for change
	Contact header is missing

	Summary of change
	Added Contact header. 

Note that a more appropriate fix might be to send px_IMS_CalleeContactUri as value of the Contact header field as described in 34.229-1, A.3.1 (though the px value is called slightly differently there – might necessitate a prose CR).

	TTCN module
	IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	Accepted in principle;
In current implementation the Contact header is included already in the 200 response for condition A1 and A3 using px_IMS_CalleeContactUri


Before change

  function f_IMS_OtherResponse_200_MessageHeaderTX(template (value) MessageHeader p_MessageHeader_Request,

                                                   template (omit) ContentType p_ContentType := omit) return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Response := cs_MessageHeader_Dummy;

    v_MessageHeader_Response.via             := p_MessageHeader_Request.via;

    v_MessageHeader_Response.fromField       := p_MessageHeader_Request.fromField;

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderTX(p_MessageHeader_Request); /* @sic R5-131897, R5-132063 - generic handling of to-tags sic@ */

    v_MessageHeader_Response.callId          := p_MessageHeader_Request.callId;

    v_MessageHeader_Response.sessionId       := p_MessageHeader_Request.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := p_MessageHeader_Request.cSeq;

    v_MessageHeader_Response.contentType     := p_ContentType;

    return v_MessageHeader_Response;

  }
After change

  function f_IMS_OtherResponse_200_MessageHeaderTX(template (value) MessageHeader p_MessageHeader_Request,

                                                   template (omit) ContentType p_ContentType := omit) return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Response := cs_MessageHeader_Dummy;

    v_MessageHeader_Response.via             := p_MessageHeader_Request.via;

    v_MessageHeader_Response.fromField       := p_MessageHeader_Request.fromField;

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderTX(p_MessageHeader_Request); /* @sic R5-131897, R5-132063 - generic handling of to-tags sic@ */

    v_MessageHeader_Response.callId          := p_MessageHeader_Request.callId;

    v_MessageHeader_Response.sessionId       := p_MessageHeader_Request.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := p_MessageHeader_Request.cSeq;

    v_MessageHeader_Response.contentType     := p_ContentType;

    v_MessageHeader_Response.contact         := p_MessageHeader_Request.contact;  

    return v_MessageHeader_Response;

  }
Change 9 – IMS_CommonDefs
	Module name
	 IMS_CommonDefs

	Reason for change
	According to RFC 4566, Section 5.14, and RFC 3550, Section 11, even-numbered ports should be used for RTP.

	Summary of change
	Introduced new constants with even numbers – the specific values might be different as long as they are picked to be even. Also we identified two places where TC 12.12 uses these media port numbers, see Changes 10 and 11. 
A more appropriate fix however might be to not introduce new constants but to just change the existing constants to even numbers. Then all users of these constants would benefit from the change (rather than just this TC). The submitter believes that such an alternative change would be correct. Note that Changes 10 and 11 would not be needed then.

	TTCN module
	Common/IMS/IMS_CommonDefs

	MCC160 Comment
	see R5s130495 Change 19


Before change

module IMS_CommonDefs {
…

  const integer tsc_IMS_MediaPort_M1 := 50061;

  const integer tsc_IMS_MediaPort_M2 := 50063;
  const integer tsc_IMS_MediaPort_M3 := 50065;
…
After change

module IMS_CommonDefs {
…

  const integer tsc_IMS_MediaPort_Mx := 50086;

  const integer tsc_IMS_MediaPort_My := 50088;

  const integer tsc_IMS_MediaPort_Mz := 50090;
…
Change 10 – cs_SDP_Media_Audio
	Template name
	cs_SDP_Media_Audio

	Reason for change
	 See Change 9

	Summary of change
	 Replaced uneven port number by even port number

	TTCN module
	 Common/IMS/IMS_SDP_Templates

	MCC160 Comment
	See Change 9


Before change

template (value) SDP_media_field cs_SDP_Media_Audio(charstring p_Fmt,

                                                    integer p_LocalPort := tsc_IMS_MediaPort_M1, 

                                                    charstring p_Transport := c_rtpAvp) :=

  /* @status    APPROVED_IMS */

    cs_SDP_Media_MultiAudio({p_Fmt}, p_LocalPort, p_Transport);
After change

template (value) SDP_media_field cs_SDP_Media_Audio(charstring p_Fmt,

                                                      integer p_LocalPort := tsc_IMS_MediaPort_Mx, 

                                                      charstring p_Transport := c_rtpAvp) :=

  /* @status    APPROVED_IMS */

    cs_SDP_Media_MultiAudio({p_Fmt}, p_LocalPort, p_Transport);
Change 11 – f_IMS_BuildSDP_AnnexC21_Step6or8
	Function name
	 f_IMS_BuildSDP_AnnexC21_Step6or8

	Reason for change
	 See Change 9

	Summary of change
	 Replaced uneven port numbers by even port numbers

	TTCN module
	 Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	See Change 9


Before change

  function f_IMS_BuildSDP_AnnexC21_Step6or8(SDP_Message p_SDP_Message_PrevOffer) runs on IMS_PTC return template (value) SDP_Message

  {

    var integer i;

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrOrigin := f_IMS_PTC_NW_Address_GetAddrStr();

    var IntegerList_Type v_IMS_MediaPortList := {tsc_IMS_MediaPort_M1, tsc_IMS_MediaPort_M2, tsc_IMS_MediaPort_M3}; 

    var template (value) SDP_Message v_SDP_Message := p_SDP_Message_PrevOffer;

    var template (value) SDP_media_desc_list v_Media_Desc_List := p_SDP_Message_PrevOffer.media_list;

    var template (value) SDP_connection v_SDP_Connection := cs_SDP_Connection(v_AddrType, v_IPAddrOrigin);

    var template (value) SDP_media_desc v_SDP_Media_Desc;

    v_SDP_Message.origin.addr_type := v_AddrType;                                     // IP address on "o=" line

    v_SDP_Message.origin.addr := v_IPAddrOrigin;                                      // IP address on "o=" line

    v_SDP_Message.connection := cs_SDP_Connection(v_AddrType, v_IPAddrOrigin);        // IP address on "c=" line

    for (i:=0; i < lengthof(v_Media_Desc_List); i:=i+1) {

      v_SDP_Media_Desc := v_Media_Desc_List[i];

      v_SDP_Media_Desc.media_field.ports.port_number := v_IMS_MediaPortList[i]; 
      v_SDP_Media_Desc.connections := { v_SDP_Connection };                           // IP address on "c=" line

      v_SDP_Media_Desc.attributes := f_SDP_ExchangeAttribute(v_SDP_Media_Desc.attributes,

                                                             cs_SDP_Attribute_curr_quos(c_remote, c_none),      // value sent by the UE in step 5/7

                                                             cs_SDP_Attribute_curr_quos(c_remote, c_sendrecv)); // ... to be replaced

      v_Media_Desc_List[i] := v_SDP_Media_Desc;

    }

    v_SDP_Message.media_list := v_Media_Desc_List;

    return v_SDP_Message;

  }
After change

  function f_IMS_BuildSDP_AnnexC21_Step6or8(SDP_Message p_SDP_Message_PrevOffer) runs on IMS_PTC return template (value) SDP_Message

  {

    var integer i;

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var charstring v_IPAddrOrigin := f_IMS_PTC_NW_Address_GetAddrStr();

    var IntegerList_Type v_IMS_MediaPortList := {tsc_IMS_MediaPort_Mx, tsc_IMS_MediaPort_My, tsc_IMS_MediaPort_Mz};
    var template (value) SDP_Message v_SDP_Message := p_SDP_Message_PrevOffer;

    var template (value) SDP_media_desc_list v_Media_Desc_List := p_SDP_Message_PrevOffer.media_list;

    var template (value) SDP_connection v_SDP_Connection := cs_SDP_Connection(v_AddrType, v_IPAddrOrigin);

    var template (value) SDP_media_desc v_SDP_Media_Desc;

    v_SDP_Message.origin.addr_type := v_AddrType;                                     // IP address on "o=" line

    v_SDP_Message.origin.addr := v_IPAddrOrigin;                                      // IP address on "o=" line

    v_SDP_Message.connection := cs_SDP_Connection(v_AddrType, v_IPAddrOrigin);        // IP address on "c=" line

    for (i:=0; i < lengthof(v_Media_Desc_List); i:=i+1) {

      v_SDP_Media_Desc := v_Media_Desc_List[i];

      v_SDP_Media_Desc.media_field.ports.port_number := v_IMS_MediaPortList[i];
      v_SDP_Media_Desc.connections := { v_SDP_Connection };                           // IP address on "c=" line

      v_SDP_Media_Desc.attributes := f_SDP_ExchangeAttribute(v_SDP_Media_Desc.attributes,

                                                             cs_SDP_Attribute_curr_quos(c_remote, c_none),      // value sent by the UE in step 5/7

                                                             cs_SDP_Attribute_curr_quos(c_remote, c_sendrecv)); // ... to be replaced

      v_Media_Desc_List[i] := v_SDP_Media_Desc;

    }

    v_SDP_Message.media_list := v_Media_Desc_List;

    return v_SDP_Message;

  }
Change 12 – f_IMS_SDP_CheckConnections
	Function name
	f_IMS_SDP_CheckConnections

	Reason for change
	The check on media level should cycle through all media descriptions rather than to repeatedly just check the first one

	Summary of change
	Replaced index 1 by i

	TTCN module
	 Common/IMS/IMS_CommonProcedures_CallControl

	MCC160 Comment
	Accepted


Before change

function f_IMS_SDP_CheckConnections(SDP_Message p_SDP_Message) runs on IMS_PTC

  {

    var boolean v_ConnectionFound := false;

    var integer i;

    if (ispresent(p_SDP_Message.connection)) {

      v_ConnectionFound := true;

    }

    if (not v_ConnectionFound) {

      // no connection specified in session description => there shall be at least one connection per media description

      v_ConnectionFound := true;

      for (i := 0; i < lengthof(p_SDP_Message.media_list); i := i + 1) {

        if (not ispresent(p_SDP_Message.media_list[1].connections)) {

          v_ConnectionFound := false;

          break;

        }

      }

    }

    if (not v_ConnectionFound) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "'Connections found neither in session description nor in media description");

    }

  }
After change

  function f_IMS_SDP_CheckConnections(SDP_Message p_SDP_Message) runs on IMS_PTC

  {

    var boolean v_ConnectionFound := false;

    var integer i;

    if (ispresent(p_SDP_Message.connection)) {

      v_ConnectionFound := true;

    }

    if (not v_ConnectionFound) {

      // no connection specified in session description => there shall be at least one connection per media description

      v_ConnectionFound := true;

      for (i := 0; i < lengthof(p_SDP_Message.media_list); i := i + 1) {

        if (not ispresent(p_SDP_Message.media_list[i].connections)) { 
          v_ConnectionFound := false;

          break;

        }

      }

    }

    if (not v_ConnectionFound) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "'Connections found neither in session description nor in media description");

    }

  }
Change 13 – cr_SDP_Attribute_fmtp
	Function name
	 cr_SDP_Attribute_fmtp

	Reason for change
	There is no core specification prescribing a blank behind the semicolon in SDP attribute fields. Specifically, the definition in RFC4566 is
attribute-fields =    *(%x61 "=" attribute CRLF)

attribute =           (att-field ":" att-value) / att-field

att-field =           token

att-value =           byte-string

The “byte-string” type does not say anything about a white-space being there or not. In RFC5576 we even find an example without a blank:
a=fmtp:98 apt=96;rtx-time=3000

So, a better assumption is to not mandate whitespace and rather leave it to implementations to accept a whitespace or not

	Summary of change
	Delete mandatory blank

	TTCN module
	Common/IMS/IMS_SDP_Templates

	MCC160 Comment
	As there seems to be no restriction on having blanks between attributes (what is acc. to the reason for change too) the implementation is changed to 
f_StringJoin(p_MediaFormatParameters, ";[ ]#(0,)")


Before change

  template (present) SDP_attribute cr_SDP_Attribute_fmtp(charstring p_FmtPattern := tsc_SDP_AnyMediaFormatDescr,

                                                         CharStringList_Type p_MediaFormatParameters := {}) :=

  { /* @status    APPROVED_IMS */

    fmtp := {

      attr_value := fl_ConcatPattern(p_FmtPattern, f_StringJoin(p_MediaFormatParameters, "; "))

    }

  };
After change

  template (present) SDP_attribute cr_SDP_Attribute_fmtp(charstring p_FmtPattern := tsc_SDP_AnyMediaFormatDescr,

                                                         CharStringList_Type p_MediaFormatParameters := {}) :=

  { /* @status    APPROVED_IMS */

    fmtp := {

      attr_value := fl_ConcatPattern(p_FmtPattern, f_StringJoin(p_MediaFormatParameters, ";")) 

    }

  };
Change 14 – f_SIP_CipheringAlgorithm_String
	Function name
	f_SIP_CipheringAlgorithm_String

	Reason for change
	According to 33.203, the ciphering string is “null” rather than “nociph” when no ciphering is used

	Summary of change
	Replaced “nociph” by “null”
Note that this issue does not necessarily occur at TC 12.12. The submitter rather used this CR to do such a general clean-up.

	TTCN module
	 Common/IMS/IMS_SIP_Templates

	MCC160 Comment
	Accepted


Before change

  function f_SIP_CipheringAlgorithm_String(IPsec_CipheringAlgorithm_Type p_CipheringAlgorithm) return charstring

  {

    var charstring v_String := "";

    select (p_CipheringAlgorithm) {

      case (des_ede3_cbc) {

        v_String := "des-ede3-cbc";

      }

      case (aes_cbc) {

        v_String := "aes-cbc";

      }

      case (nociph) {

        v_String := "nociph";

      }

    }

    return v_String;

  }
After change

  function f_SIP_CipheringAlgorithm_String(IPsec_CipheringAlgorithm_Type p_CipheringAlgorithm) return charstring

  {

    var charstring v_String := "";

    select (p_CipheringAlgorithm) {

      case (des_ede3_cbc) {

        v_String := "des-ede3-cbc";

      }

      case (aes_cbc) {

        v_String := "aes-cbc";

      }

      case (nociph) {

      v_String := "null"; 
      }

    }

    return v_String;

  }
Change 15 – f_SIP_SipUrl2Str
	Function name
	 f_SIP_SipUrl2Str

	Reason for change
	When TTCN receives an IPv6 address (e.g. in To or From headers), decoded it and then in transforms it back into a string via f_SIP_SipUrl2Str it needs to add brackets acting as delimiters for IPv6 addresses

	Summary of change
	Added brackets

	TTCN module
	Common/IMS/IMS_SIP_Templates

	MCC160 Comment
	Accepted in principle


Before change

  function f_SIP_SipUrl2Str(SipUrl p_SipUrl) return charstring

  { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

    var SipUriComponents v_SipUriComponents;

    var charstring v_String := p_SipUrl.scheme & ":";

    select (p_SipUrl.scheme) {

      case (c_sipScheme) {

        v_SipUriComponents := p_SipUrl.components.sip;

        if (ispresent(v_SipUriComponents.userInfo)) {

          v_String := v_String & v_SipUriComponents.userInfo.userOrTelephoneSubscriber & "@" ; // @sic R5s120530 change 1.3 sic@

        }

        v_String := v_String & v_SipUriComponents.hostPort.host; // @sic R5s120530 change 1.3 sic@
        if (ispresent(v_SipUriComponents.hostPort.portField)) {

          v_String := v_String & ":" & int2str(v_SipUriComponents.hostPort.portField);

        }

      }

      case (c_telScheme) {

        v_String := v_String & p_SipUrl.components.tel.subscriber;

      }

      case (c_urnScheme) {

        v_String := v_String & p_SipUrl.components.urn.namespaceId & p_SipUrl.components.urn.namespaceSpecificString;

      }

      case else {

        v_String := v_String & p_SipUrl.components.other;

      }

    }

    return v_String;

  }
After change

function f_SIP_SipUrl2Str(SipUrl p_SipUrl) return charstring

  { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

    var SipUriComponents v_SipUriComponents;

    var charstring v_String := p_SipUrl.scheme & ":";

    select (p_SipUrl.scheme) {

      case (c_sipScheme) {

        v_SipUriComponents := p_SipUrl.components.sip;

        if (ispresent(v_SipUriComponents.userInfo)) {

          v_String := v_String & v_SipUriComponents.userInfo.userOrTelephoneSubscriber & "@" ; // @sic R5s120530 change 1.3 sic@

        }

        if(f_IpAddressIsIPv6( v_SipUriComponents.hostPort.host))                      
        {                                                                             
            v_String := v_String & "[" & v_SipUriComponents.hostPort.host & "]";      
        }                                                                             
        else                                                                          
        {   

        v_String := v_String & v_SipUriComponents.hostPort.host; // @sic R5s120530 change 1.3 sic@

         }
        if (ispresent(v_SipUriComponents.hostPort.portField)) {

          v_String := v_String & ":" & int2str(v_SipUriComponents.hostPort.portField);

        }

      }

      case (c_telScheme) {

        v_String := v_String & p_SipUrl.components.tel.subscriber;

      }

      case (c_urnScheme) {

        v_String := v_String & p_SipUrl.components.urn.namespaceId & p_SipUrl.components.urn.namespaceSpecificString;

      }

      case else {

        v_String := v_String & p_SipUrl.components.other;

      }

    }

    return v_String;

  }

